

# CEEMAC

## A Visual Composition System for Apple II computers


Reviewed by Andre Schklowsky

CEEMAC is a language for creating dynamic, interactive, animated and abstract visual compositions called "scores." This unique software package already has a highly involved group of users. The unprotected floppy comes with a substantial (70 packed pages) manual and a twelve page tutorial.

I discovered CEEMAC in November, 1981, at the Small Computers In The Arts Conference in Philadelphia. Actually, it hadn't become CEEMAC at that time. During the conference breaks, Eric Podietz, one of the conference's organizers, would boot the CEEMAC demo disk, called "Vagabondo," and we would all sit back in awe, asking, "How can you do this?" We were treated to a little recital of very fast, animated video. What we were witnessing was the expression of a new art form that was unique to personal computers and especially the Apple, with its marvelous graphic capabilities. Vagabondo was free (and still is) for copying. So I made myself a copy. A few months later CEEMAC was advertised, and I became, as I later learned, one of the first 100 addicts.


When CEEMAC is booted, three distinct programs are loaded in memory: the editor, the proof reader and the interpreter. The fourth part is missing. That is the DOS interface — more on that later.

**By Brooke W. Boering (Vagabondo Enterprises, 1300 E. Algonquin — 3G, Schaumburg, IL 60195.) System requirements: 48K Apple II with disk drive and game paddles. Prices: CEEMAC \$75; Maestro \$45; Vagabondo (demo disk) free to copy at your Apple users group.**



Diamond Dazzler by Brooke W. Boering.


```
SCORE: DIAMOND DAZZLER
  CLEAR [0;0]
  SETASP [2;0]
  V1 = 0
  SPEED [0;0]
  F
  XY1 = $B0;RANDOM
  XY2 = Y1;$B0
  V1 = V1 + 1
  FOR $20
    BLINE [0;7]
    Y1 = Y1 - V1
    X2 = X2 + V1
  BRASE [$20;0]
  AGAIN
  CEEMAC REL 1.1
```



Take-Off by Andre Schklowsky.


```
SCORE: TAKE OFF
  1
  CLEAR [0;0]
  SPEED [FF;FF]
  V1 = $B0
  V2 = $B0
  V3 = $FF
  V4 = $B0
  FORCE1 = 2:17
  DENSTY = $FF
  COLOR = $FF
  SKY [$B0;0]
  FORCE2 = 11:29
  DO TIL V2 > $E0
    X1 = V1
    Y1 = V2
    X2 = V3
    Y2 = V4
    SPLINE [0;2]
    PULSE [0;1]
    SPRASE [8;0]
    V2 = V2 + 5
    V3 = V3 - 5
  AGAIN
  GOTO 1
  CEEMAC REL 1.1
```






```
SCORE: SPLINES#2
SPEED [0; $FF]
CLEAR [0; 0]
X1 = $80
Y1 = $80
FORCE2 = 5:15
FORCE1 = 8:31
DENSITY = $FF
COLOR = $FF
FOR $20
    Y2 = LIST1 , X2
: NOTE SYMMETRY OF 3
SPLINE [0; 3]
X2 = X2 + 7
AGAIN
A
GOTO A
CEEMAC REL 1.1
```

Splines #2 by Andre Schlowsky.



```
SCORE: SPLINES#4
1
CLEAR [0; 0]
SPEED [$FF; $FF]
V1 = $80
V2 = $80
V3 = $FF
V4 = $80
FORCE1 = 2:17
DENSITY = $FF
COLOR = $FF
SKY [0; 0]
FORCE2 = 3:45
DO TIL 1 = 3
    X1 = V1
    Y1 = V2
    X2 = V3
    Y2 = V4
SPLINE [0; 9]
SPRASE [$40; 0]
V2 = V2 + 5
V3 = V3 - 5
AGAIN
GOTO 1
CEEMAC REL 1.1
```

Splines #4 by Andre Schlowsky.



```
SCORE: SPLINES/JACKSON
CLEAR [0; 0]
SPEED [$FF; 0]
XY1 = 0; 0
V1 = V1 - 5
V2 = V2 + 1
V4 = 7
DENSITY = $80
FOR $FF
    V1 = V1 - 1
    V2 = V2 + 1
FORCE1 = V2
FORCE2 = V1
XY0 = X1; Y1
X0 = X0 % 5
Y0 = Y0 % $17
X1 = X1 % 4
Y2 = Y2 % 4
SPLINE [0; $0F]
SPRASE [$40; 0]
AGAIN
CEEMAC REL 1.1
```

CEEMAC, *continued*

## Commands and Error Traps

The editor and proof reader actually are one program from the user's point of view. It is a well-designed line editor for creating animated graphics "scores." You need to remember very few keys, and none of the editing keys are control keys. Once you have used it a few times you remember it for quite a while. The proof reader is there when you need it — if you enter an incorrect instruction, a bell rings to force you to correct the syntax. This editor also is very forgiving. For instance, if you enter `v1 + 1`, it will translate it into `V1 = V1 + 1`. When a macro requires two parameters, and you enter only the first, it enters a default for the second one, and establishes the correct syntax itself. The choice for the keys to move up and down in the "score" is judicious: the hyphen to go up, the RETURN to go down. They are closest to the repeat key. To insert a line, press I. To delete one, press D.

Pressing CTRL A switches from editor to interpreter. Testing is easy because the editor remembers where it left the cursor so you can make small changes and try them out immediately.

Some errors are not syntax errors, and the proof reader may not spot them. If you want to GOTO a non-existent label, the proof reader won't warn you in advance. But when the interpreter fails to find the label, it sends you back to the editor and prints an error message at the bottom of the screen. Actually, few programming errors are left when the program starts, since the proof reader catches most of them at entry time. This is one of the most important lessons we have learned (starting with on-line time-sharing systems). When less time elapses between writing a faulty instruction and knowing about it, you can correct it more easily and proceed. The old timers may remember batch jobs with their two or three hours turn-around time — not the easiest way to debug a program. This kind of immediacy, learned from video games, is one of the most exciting aspects of personal computers and will have a considerable impact on how we will learn in the future.

## CEEMAC Features

Look at some of the scores I have included in this review, and you will notice that this language looks more like Pascal than BASIC. CEEMAC offers a rich choice for sequence controls, such as:



- For x1
- Skip if x1 > x2
- Do unless v1 < v2
- Exit if x1 odd
- Again while x1 < \$80
- Gosub chc (goto subroutine chc)

The choice of mathematical and logical operators is impressive. In addition to the four arithmetical functions, you have "mod", "and", "or", "eor" and the weird but powerful "%." It is positive until it encounters overflow. Then it reverses to negative until underflow. This is useful when you want to control bouncing at the edges of the screen. Another weird one is "+ or -", which provides a random "+" or a "-". A few math functions are missing, particularly the trigonometric functions, but the tables included take care of these.

CEEMAC has many variable types, such as:

- Free variables: v1,...vf
- Variables for coordinates: x0,y0,...x3,y3 can be combined, as in xy0 \$20 40, a short cut for x0 = \$20, y0 = \$40

● Random variables: The palette of randomly generated variables is one of CEEMAC'S outstanding features.

● "Finger" variables allow the program to get input from the keyboard, the paddles or the push buttons.

Macros are the built-in routines for creating the visual display with a minimum of programming. BASIC, for instance, has four such macros: HGR (clear screen), HPLLOT (draw a line or just a dot), DRAW (draw a shape), and XDRAW (erase a shape). CEEMAC has CLEAR, BLINE, ADOT and SHAPE, in addition to the following:

- ABOX (to draw a four-sided figure, filled or not)
- SPLINE (to draw a Jordan curve). This is my favorite drawing macro. I frequently use it with different symmetry codes. It produces a more organic kind of display that most people do not associate with computers.
- SKY (fill the screen with stars)
- GRID
- PULSE uses the Apple color scheme to shift the screen color phase — a nice effect.

In addition to these macros, CEEMAC has several utilities. A symmetry parameter gives a choice of twelve different symmetries. On a figure, imagine drawing a dot (or a line, spline or shape) at a #, and without any further programming getting the same dot at all the other #s. Some macros have an automatic stack erase feature. For instance, BRASE 8 erases the eight latest lines created with BLINE. Similarly, SPRASE erases the n oldest splines. This feature is essential when you consider the dynamic quality of your display. Without it the screen would soon fill up completely.

In addition to the graphic macros, you may use:

- SWAP to exchange the content of two variables.
- SPEED to alter the speed of execution.
- TRACE to print the value of a given variable.
- PAUSE to wait for a keypress.
- STEP (trace plus pause) for debugging.



## SOFTSIDE ORDERING INFORMATION

### FORM OF PAYMENT

#### USA

VISA, MasterCard, certified checks, money orders and personal checks are accepted.

#### Canada/Mexico

The preferred method of payment is by VISA or MasterCard. A bank check is acceptable if it has been preprinted for payment in U.S. dollars. No personal or company checks accepted.

#### Other Foreign Orders

Payment must either be by a bank check drawn on a U.S. bank payable in U.S. dollars or by affiliated bank credit cards of VISA or MasterCard.

### GUARANTEE

All software is guaranteed to load and run. If you experience difficulties with the product within 30 days, it may be returned for replacement. Send your properly protected tape or disk to the attention of the Customer Service Representative and include your name, address, and the reason it is being returned.

### LIABILITY

All software is sold on an as-is basis. SoftSide assumes no liability for loss or damage caused or alleged to be caused directly or indirectly by products sold or exchanged by them or their distributors, including, but not limited to, any interruption in service, loss of business or

anticipatory profits or consequential damages resulting from use or operation of such software.

### PRICES

Prices are subject to change without notice. We are not responsible for typographical errors.

Unless otherwise noted in a published advertisement, the following prices are in effect as of this issue:

	USA/Canada APO/FPO	USA/Canada FIRST CLASS Mexico	Other Foreign
SoftSide Magazine* (yr)	\$30	\$40	\$62
SoftSide Magazine* (6 mo.)	\$18	\$23	\$34

	USA APO/FPO	Mexico Canada	Other Foreign
CV (year) & magazine (6 mo.)	\$99 \$55	\$119 \$65	\$149 \$80
DV (year) & magazine (6 mo.)	\$149 \$79	\$169 \$89	\$199 \$104


\*Includes SoftSide Selections.

### BACK ISSUES

Minimum order — \$10. Price includes shipping to the 48 states only. Alaska, Hawaii, Puerto Rico, APO/FPO, and ALL foreign orders — postage is additional.

ALL Foreign orders and all magazine/media combination orders — Order directly from SoftSide, 10 Northern Blvd., Northwood Executive Park, Amherst NH 03031.

## CEEMAC, continued





Key 4 on the Vagabondo demo.

```
SCORE: SPLINES#3
  SPEED [0;$FF]
CLEAR [0;0]
X1 = $B0
Y1 = $B0
FORCE2 = 5:15
FORCE1 = 8:31
DENSITY = $FF
COLOR = $FF
FOR $20
  Y2 = LIST1 , X2
  : NOTE SYMMETRY OF 5
  SPLINE [0;5]
  X2 = X2 + 7
  AGAIN
A
GOTO A
CEEMAC REL 1.1
```

```
SCORE: K4-ICE
:
: INITIALIZE
:
: CLEAR [0;0]
: SPEED [0;0]
: COLOR = $FF
:      FF = WHITE
:
V1 = $10
: V1 IS LOOP FOR PATTERN DISPLAY
:   NOTE: $10 IS HALF OF THE
:         ERASE DELAY.
:   THIS WILL GIVE TWO
:   PATTERNS WITH THIRD
:   BEING DRAWN AS FIRST
:   IS BEING ERASED.
:
V4 = 7
: V4 SETS SYMMETRY
:   NOTE: 7 GIVES MAXIMUM MOVEMENT
:         ON THE SCREEN.
:
F
: SET PATTERN VARIABLES
:
XY1 = RANDOM;RANDOM
X2 = $90
: NOTE: 90 FORCES OFFSET OF CENTER
:
: DISPLAY PATTERN; ERASE PRIOR
FOR V1
  BLINE [0;V4]
  BRASE [$20;0]
  Y2 = Y2 + 8
  : SEE FOOT 1
  AGAIN
:
: FOOT 1:
:   CONTINUOUS INCR. OF Y2
:   WILL CAUSE WRAPAROUND OF
:   PATTERN AT SCREEN LIMIT,
:   THUS A MORE COMPLEX IMAGE.
:
CEEMAC REL 1.1
```

Splines #3 by Andre Schklowsky.



## Shapes

The shape facility is one of the most intriguing and frustrating in CEEMAC. It is rich in possibilities, and using the supplied shapes is fairly easy. Unfortunately, creating new shapes is very difficult. CEEMAC comes with several shape tables, but only one is loaded. Using them is easy, once you have learned to match a code with the corresponding shape, and to realize that drawing a shape depends on a number of parameters. To use a shape you have to set:

- ROTATE (scaled on a 256 point compass).
- SIZE (one of eight sizes).
- XSCALE and YSCALE (horizontal and vertical multipliers).

## Lists and Tables

A list is a collection of 256 constants brought in from the disk. Two of these tables are loaded with CEEMAC. You may create more tables on your own and load up to four of them in memory. The LISTn variable gives you access to one of the elements of the nth table. For instance, in the expression x1 = list1, v1 will return in x1 the v1th element of the first list. With some basic programming you could create these tables and since the CEEMAC floppy is not protected, you may modify the Hello program to load your own tables.

## Additional Features

Printing your scores is simple: While you are in the editor, just press shift-1 (for printer in slot 1), shift 5 (for slot 5).

Through the cassette "in" port you may "read" music and synchronize your "visual" score with the incoming pitch. You can use the ANOTE macro to create a sound on the Apple speaker and to the cassette "out" port. I met a CEEMAC user who used this to synchronize the frame advance of an 8mm camera to film some of his scores.

You may add text to your displays through the use of a shape table provided on the CEEMAC floppy (but not loaded by the Hello program). I altered the Hello program to use the "t-shape" table in a score that translated any key to a letter shape. My three-year-old son used this program, called "Big Letters," to become familiar with the keyboard. He could change the size of the letters with one paddle and their orientation with the second one.

Throughout CEEMAC, the default numerical notation is hexadecimal. You may use decimal as input but it will be translated into hex. This may turn off some users. In fact, it is just a convention, and since the numbers go only from 0 to 255 (sorry — 0 to \$FF), the hardship is minimal.

CEEMAC's DOS interface is limited. An Apple "hacker" accustomed to BLOADs, RENAMEs and BSAVEs will have no problem. To play a score, simply BLOAD it and call 2048. To create a new one, use an old one. First delete all its lines, write the new score, try it out, and amend it until you are satisfied. Then press CTRL/C to exit from CEEMAC, and BSAVE the score you just created. You have to reenter that line from the keyboard; or, as a hacker would, use the ESC J and I and the arrow keys to overwrite the line and issue the command. Imagine what a beginner goes through to learn CEEMAC's *modus operandi*. It may not be everyone's cup of tea, but CEEMAC was designed more for the programmer with a flair for creating exciting visual displays, than for the non-computer artist who wants to learn a programming language and become a "score" composer.

## Saving HIRES Pictures

Very often, as you sit back and watch a score, you may want to save the picture on a floppy to show as a fixed image at a later date. Theoretically it is very simple:

- Stop the score by pressing the space bar;
- Press CTRL/A, then CTRL/C;

- BSAVE picture name, A\$2000, L\$1fff.

In the first release of CEEMAC you had to contend with a memory conflict between where it saves the score and where it creates the picture. In the latest release the problem is solved. Thank you, Mr. Boering!

## Documentation

The personal computer is a versatile tool for creative people who want to bridge the gap between the arts and science. But, to use and enjoy it, you must learn some of the conventions of operating the Apple II. With CEEMAC, Brooke Boering has pushed the Apple II to some of its limits. To see what he does with the next generation of HIRES machines, based on Motorola's 68000 16-bit microprocessor and NEC's 7220 graphics chip, will be interesting. For the time being, I'll enjoy CEEMAC on my faithful Apple II.

The 70-page manual, and the twelve-page tutorial are thorough, although I don't like the exclusive use of upper case letters. The index is very useful. I have not used the tutorial to get acquainted with CEEMAC because it did not exist for the original version. It looks well-written and should get most people started. The CEEMAC floppy contains many sample scores. Newcomers and old hands should read them, modify them, and of course, show them.

## Utilities For CEEMAC

I have been using computers as a programmer since 1967 and have used many computer languages, from machine language to a fourth generation language called Admins, and since I own an Apple II, BASIC, Logo, GraForth and 6502 assembly language. CEEMAC is another one. Like any other language, you have to use it to learn it, and the author insists on the learning-by-trying aspect in the tutorial. When I purchased CEEMAC it did not have a tutorial, so I decided to read one of the "scores" and compare it line-by-line with the manual to understand the instructions. I modified the score in various ways to see what would happen, learning a great deal about CEEMAC in the process.

With *Maestro*, the visual album maker, Boering and Vagabondo have given us not only a tool to create individual scores, but also a means to package them together. After a few months of playing with CEEMAC, you

may have become an addict, and want to create an album of your own. To show your CEEMAC work, you must keep switching between the program and DOS, and issue many catalog and BLOAD commands. I mentioned earlier that the Vagabondo demo disk is a collection of scores, each of which can be executed by pressing a key. *Maestro* is a utility program to read one or more of your CEEMAC floppies and help you compose an album similar to the demo disk. It is designed for experienced CEEMAC users, who already have a collection of scores to package.

*Maestro* works well, provided you do some preparation before starting.

- Review your CEEMAC scores carefully before you start if you want to skip the "work-in-progress" scores.
- Have your *Maestro* and CEEMAC data floppies handy. The first one must have at least one list table and one shape table.
- Make a copy of *Maestro's* second side. The program will record the album on this disk.

As *Maestro* goes through your scores, write down the letter associated with each one. That helps if you want to modify subsequent versions of this album. Several albums can fit on one floppy. Unfortunately, here again the DOS interface is limited. *Maestro* recognizes only the file names it expects: one for the scores, one for the comments and one for the HIRES picture (the album cover). Unless you want to do a lot of last minute RENAMEs, keep it simple and stick to the "one floppy/one album" scheme.

The documentation is not as thorough as CEEMAC's. The assumption may be that this product is directed toward the CEEMAC aficionado, who will get the idea quickly and work out the problems on his own.

## CEEMAC Newsletter

Independently of Vagabondo Enterprises, Ken Sherwood publishes (a few times a year) a newsletter full of information about CEEMAC and related subjects. The focus is on scores, and it has become a forum for CEEMAC users to share their work. Some of the scores shown here come from, or are adapted from, scores in the newsletter. To subscribe, write to: Ken Sherwood 117 N. 25th St. Reading, PA 19606. Subscriptions are \$15 a year.